



Don't Be a SAS® Dinosaur: Modernize Your SAS Programs

by Warren Repole

Arrays: Temporary Elements


Scenario:

You want to perform a table lookup using an array of constants.

You can create the array by assigning initial values to the elements, but the underlying variables appear in the output data set by default.

The old way: Drop the Lookup Array Elements

The original approach is to eliminate the array elements from the output data set using the DROP statement or the DROP= data set option.



```
data Differences(drop=i);
  set sashelp.class;
  array BaseLine {3} (13.32 62.33 100.03);
  drop BaseLine1-BaseLine3 ;
  array Actuals {3} Age Height Weight;
  array Diffs {3} DiffAge DiffHgt DiffWgt;
  do i = 1 to 3 ;
    Diffs{i} = Actuals{i} - BaseLine{i};
  end;
run;
proc print data=Differences;
  title1 "Differences from BaseLine";
run;
```

Please direct all correspondence to:

Warren Repole, 1705 Palm Springs Dr, Vienna VA 22182-2331

sasdinosaur@repole.com

www.repole.com/dinosaur

The new way: Create the Array with Temporary Elements (available in SAS Version 6)

An alternate approach is to specify the array to contain temporary elements.

```
data Differences(drop=i);
  set sashelp.class;
  array BaseLine {3} _temporary_ (13.32 62.33 100.03);
  array Actuals {3} Age Height Weight;
  array Diffs {3} DiffAge DiffHgt DiffWgt;
  do i = 1 to 3 ;
    Diffs{i} = Actuals{i} - BaseLine{i};
  end;
run;
proc print data=Differences;
  title "Differences from BaseLine";
run;
```

Advantages of the alternate approach:

- No additional statements or data set options are needed in order to prevent the array elements from appearing in the output data set.

Disadvantages of the alternate approach:

- The number of elements must be declared explicitly in the ARRAY statement. [Lower and upper bounds are permitted.]

Additional documentation for this technique can be found in *SAS® 9.2 Language Reference: Dictionary*. Cary, NC: SAS Institute Inc.

Visit <http://support.sas.com/documentation/onlinedoc/sas9doc.html> for SAS 9 documentation.

Go to <http://www.repole.com/dinosaur/temparray.html> for the sample code and output for this topic.

These techniques are mentioned in other SAS references and publications:

SAS Sample 24773 (<http://support.sas.com/kb/24/773.html>)

SUGI Paper 158-29 (<http://www2.sas.com/proceedings/sugi29/158-29.pdf>)

Precambrian Iron Formations; <http://www.stat.tamu.edu/~twehrly/601/examples/iron.pdf>.